


פתרון המבחן (חלקי)

שאלה 1 (75 נקודות)

תכנן FIFO (רכיב זיכרון הפועל לפי המשטר First In First Out המכונה גם טור) המתווך בין שני מערכות הפועלות בקצבים שונים. ה-FIFO מוצג באיור 1.



משמעות של כל פורט של FIFO מפורטת בטבלה הבאה:


Write side		Read side	
wr_clk	Write clock	rd_clock	Read clock
wr_req	Write request	rd_req	Read request
ff	Fifo full flag	fe	Fifo empty flag
wr_data	Write data	rd_data	Read data

במימוש ה-FIFO יש להשתמש בטיפוס מערך לייצוג זיכרון הפנימי של ה-FIFO כאשר גודל המילה בזיכרון (FIFO_WIDTH) ועומק ה-FIFO (FIFO_DETPH) קרי מספר המילים המרבי היכול להימצא ב-FIFO נקבעים ע"י **generic** וה-FIFO_DETPH הוא חיזוקה שלמה של 2. אין למנוע (בלוגיקה של ה-FIFO) כתיבה ל-FIFO במצב **ff** וקריאה מה-FIFO במצב **fe**. יש להתייחס לעובדה שה-FIFO משני צדדיו מנוהל ע"י **שעונים שונים** !!!, דבר המשפיע על עדכון ה-**ff** וה-**fe**. על מנת לממש לוגיקה של ה-full וה-empty יש לנהל בנוסף למונים המשמשים ככתובות לכתובה **wr_pointer** (בצד הכתיבה) והקריאה **rd_pointer** (בצד הקריאה) עוד שני מונים ב-Gray Code העוברים בהצלבה מצד לצד. מונים אלה נדגמים פעמיים בכל צד לפי השעון שלו ורק אז מתבצעת בדיקה הקובעת את ה-**ff** וה-**fe**.

המשימה: ממש מערכת הנ"ל בשלבים הבאים:

1. שרטט דיאגרמת בלוקים מפורטת של המערכת (10 נקודות)
2. כתוב קוד בר-סינתזה של המערכת כולה ב-VHDL (40 נקודות)
3. כתוב Test-Bench הבודק את ה-FIFO, הבדיקה חייבת להתייחס למצבי ה-**ff** וה-**fe** (25 נקודות)

דיאגרמת בלוקים של ה-FIFO



```

1> library ieee;
2> use ieee.std_logic_1164.all;
3> use ieee.std_logic_arith.all;
4> use ieee.std_logic_unsigned.all;
5>
6> entity dp_fifo is
7>   generic (
8>     FIFO_WIDTH: integer; FIFO_DEPTH_LOG2: integer);
9>   port (
10>     rst      : in  std_logic;
11>     -- write side
12>     wr_clk   : in  std_logic;
13>     wr_data  : in  std_logic_vector((FIFO_WIDTH-1) downto 0);
14>     wr_req   : in  std_logic;
15>     ff      : out boolean;
16>     -- read side
17>     rd_clk   : in  std_logic;
18>     rd_data  : out std_logic_vector((FIFO_WIDTH-1) downto 0);
19>     rd_req   : in  std_logic;
20>     fe      : out boolean);
21> end entity dp_fifo;
22>
23>
24> architecture arc_dp_fifo of dp_fifo is
25>   type ram is array ((2**FIFO_DEPTH_LOG2)-1) downto 0 of std_logic_vector(wr_data'range);
26>   signal fifo_buff : ram;
27>   signal wp,wp_gray,wp_gray_s0,wp_gray_s1 : std_logic_vector(FIFO_DEPTH_LOG2 downto 0);
28>   signal rp,rp_gray,rp_gray_s0,rp_gray_s1 : std_logic_vector(FIFO_DEPTH_LOG2 downto 0);
29>   alias wr_addr : std_logic_vector((FIFO_DEPTH_LOG2-1)downto 0) is wp((FIFO_DEPTH_LOG2-1) downto 0);
30>   alias rd_addr : std_logic_vector((FIFO_DEPTH_LOG2-1)downto 0) is rp((FIFO_DEPTH_LOG2-1) downto 0);
31> begin
32>   wr_side: process (wr_clk, rst) is
33>     variable bcnt, nextb : std_logic_vector(FIFO_DEPTH_LOG2 downto 0);
34>   begin
35>     if (rst = '0') then
36>       wp<=(others => '0');
37>       wp_gray<=(others => '0');
38>       rp_gray_s0<=(others => '0');
39>       rp_gray_s1<=(others => '0');
40>     elsif rising_edge(wr_clk) then
41>       if (wr_req='1') then
42>         fifo_buff(conv_integer(wr_addr))<=wr_data;
43>         wp<=wp+1;
44>       end if;
45>       for i in bcnt'range loop
46>         if (i=bcnt'high) then
47>           bcnt(i):=wp_gray(i);
48>         else
49>           bcnt(i):=wp_gray(i) xor bcnt(i+1);
50>         end if;
51>       end loop;
52>       nextb:=bcnt+wr_req;
53>       wp_gray<=nextb xor ('0' & nextb(nextb'left downto 1));
54>       rp_gray_s0<=rp_gray;
55>       rp_gray_s1<=rp_gray_s0;
56>     end if;
57>   end process wr_side;
58>
59>   rd_side: process (rd_clk, rst) is
60>     variable bcnt, nextb : std_logic_vector(FIFO_DEPTH_LOG2 downto 0);
61>   begin
62>     if (rst = '0') then
63>       rp<=(others => '0');
64>       rp_gray<=(others => '0');
65>       wp_gray_s0<=(others => '0');
66>       wp_gray_s1<=(others => '0');
67>       rd_data<=(others => '0');
68>     elsif rising_edge(rd_clk) then
69>       if (rd_req='1') then
70>         rd_data<=fifo_buff(conv_integer(rd_addr));
71>         rp<=rp+1;
72>       end if;
73>       for i in bcnt'range loop
74>         if (i=bcnt'high) then
75>           bcnt(i):=rp_gray(i);
76>         else
77>           bcnt(i):=rp_gray(i) xor bcnt(i+1);
78>         end if;
79>       end loop;
80>       nextb:=bcnt+rd_req;
81>       rp_gray<=nextb xor ('0' & nextb(nextb'left downto 1));
82>       wp_gray_s0<=wp_gray;
83>       wp_gray_s1<=wp_gray_s0;
84>     end if;
85>   end process rd_side;
86>
87>   ff<= (wp_gray(wp_gray'left)/=rp_gray_s1(rp_gray_s1'left)) and
88>        (wp_gray((wp_gray'left-1) downto 0)=rp_gray_s1((rp_gray_s1'left-1) downto 0));
89>   fe<= (rp_gray=wp_gray_s1);
90> end architecture arc_dp_fifo;

```

שאלה 2 (25 נקודות)

יש לכתוב שתי פונקציות הבאות:

1. (15 נקודות) פונקציה המקבלת מערך של שלושה `std_logic_vector(7 downto 0)` ומחזירה

תוצאה מטיפוס `std_logic_vector`. הפונקציה מחשבת חציון (median) של המערך הנתון, כלומר ערך שהוא גדול או שווה לערך המזערי במערך ושווה או קטן מהערך המרבי במערך. פונקציה זו משמשת כפונקציה עזר לפונקציה אם מסעיף הבא.

2. (10 נקודות) פונקציה אשר מחשבת ומחזירה (באמצעות הטיפוס `std_logic_vector`) חציון במערך שאלמנט בסיסי שבו הוא מערך מסעיף קודם, כלומר הארגומנט של הפונקציה הוא מערך של מערכים (מטריצה).

על מנת לממש פונקציות הנ"ל, תחילה יש להגדיר טיפוס מסוג מערך בעבור כל פונקציה בהתאם למוסבר לעיל. במימוש הפונקציות אין להגביל ארגומנטים, כלומר אין לקבוע מה הוא ה-`range` של הארגומנט, הפונקציות אמורות לקרוא מהארגומנט את המאפיינים (attributes) שלו ולהשתמש בהם.

```
type curr_color is array (natural range <>) of bit_vector(7 downto 0);
type matrix2d is array (0 to 2) of curr_color(0 to 2);
```

```
function padding (arg : matrix2d) return bit_vector is
    variable buff : curr_color(0 to 2);
    function median(d : curr_color(0 to 2)) return bit_vector is
        variable temp : bit_vector(7 downto 0);
    begin
        if (d(2) >= d(1) and d(2) <= d(0)) then
            temp:= d(2);
        elsif (d(2) >= d(0) and d(2) <= d(1)) then
            temp:= d(2);
        elsif (d(1) >= d(0) and d(1) <= d(2)) then
            temp:= d(1);
        elsif (d(1) >= d(2) and d(1) <= d(0)) then
            temp:= d(1);
        elsif (d(0) >= d(1) and d(0) <= d(2)) then
            temp:= d(0);
        elsif (d(0) >= d(2) and d(0) <= d(1)) then
            temp:= d(0);
        else
            temp:= d(0);
        end if;
        return temp;
    end function median;
begin
    for i in arg'range loop
        buff(i) := median(arg(i));
    end loop;
    return median(buff);
end function padding;
```